
Model railroad crossing gate with an 
ATtiny24 and a servomotor
If you are

• addicted to railway models, AND you are

• not among the richest to invest hundreds of dollars to just open and close model 
railway gates smoothly (with this here available for less than 10$ or € all in all), OR 
if you are

• a complete newbie in AVR assembly programming and want to learn 16-by-8 bit bi
nary multiplication and 16-by-8 bit binary division in practice, OR if you are 

• bored by C coding, because it gives you no real control over the hardware of the 
controller and you do not find a library that does exactly what you need,, OR if you 
are

• disappointed about C libraries because they reduce you to a dump caller of routines 
that you neither understand nor have been written by yourself  nor you have a 
chance to fully understand them nor have any control over, OR if you are

• completely on a mass drug called Arduino, your projects were so far solely on an 
ATmega328 and you don't know something else beyond that, OR if you are 

• searching for new worlds that open up your horizon of deeper understanding oppor
tunities and limits of modern technology,

this is a small and simple to build project that fits your needs. The whole software is docu
mented here, so you can even understand how it works. If necessary you can inspect it in
struction word by instruction word, no hidden things or in-transparent secrets here. And it 
provides you with a very flexible and easy to adjust piece of lean, stand-alone hardware.

1 Properties
Here is a servomoter controller for model railroad crossing gates with the following prop
erties: 

• opens and closes the gate with two push buttons, 
• allows you to adjust the upper and lower positioning with two trim potentiometers, 
• allows you to adjust the speed of movement with a separate trim potentiometer, 
• displays the current state with a red/green duo-LED: 

• Gate fully open: permanently green, 
• Gate fully closed: permanently red, 
• Gate opens or closes: LED blinks red,

• wide operating voltage range, 2.7 to 5.5 Volt, depending from your servomotor, 
• Open source assembler code, available for download. 

1 Hardware

1.1 Schematic

The schematic is simple. 
The  controller  ATtiny24 
controls it all. Three 10k 
trim  potentiometers  are 
attached  to  the  ADC0, 
ADC1  and  ADC2  pins. 
The  red/green  duo  LED 
is  attached  to  the  port 
pins  PB0  (anode  red, 
cathode green) and PB1 
(cathode  red,  anode 
green).  The  two  push 
buttons are on the INT0 
and  PCINT7  input  pins. 
The  pulse-width-modu
lated signal for the ser
vomotor is generated on 
output OC1A. If programming of the ATtiny24 shall be performed a 6-pin In-System-Pro
grammer can be attached via the standard ISP6 plug. 

1.2 Electrical functions

All electrical functions are performed by the micro-controller ATtiny24. 

• He runs with the internal RC clock generator of 8 MHz, divided by 8. This is the 
standard configuration with which the ATtiny24 is shipped. 

• The angle, at which the three trim potentiometers on the pins 13, 12 and 11 of the 
ATtiny24 are actually trimmed, is read by the analog-to-digital converter. 64 mea
surements are summed up for each of the three channels. Such a cycle lasts one 
third of a second. If the gate is in the open position, the gate follows any changes 
on the open trim. If in close position, the gate follows changes on the close trim. 

• Closing of the gate is initiated by the push button on the INT0 input pin, and if the 
INT0 pin is grounded during start-up. 

• Opening of the gate is initiated by the push button on the PCINT7 input pin and by 
default on start-up. 

• The PWM output signal that controls the position of the servomotor is generated on 
output pin OC1A. The pulse width of the active-high signal determines the angle, to 
which the servomotor moves. 50 PWM cycles are performed per second. 

• The micro-controller can be programmed via the ISP6 interface. Programmer pulses 
are received/send via the MOSI, MISO and SCK pins, programming is initiated by 
pulling the RESET input low. Programming does not interfere with the servomotor 
because the programming pulses are much faster than servomotor signals. 



1.3 PWM signal for the servomotor

The  PWM signal  for  the  servo
motor looks like this: the pin is 
on for  a certain  period and off 
for  the  remaining  duration  to 
20 ms. 

The active signal duration determines the posi
tion of the motor. It can be between 900 and 
2,100 µs  length.  The  shortest  duration  brings 
the motor to its rightmost position, the longest 
to its leftmost position. 1,500 µs is the middle 
position. The position of the trim potentiometer 
P1 varies the signal duration from 1,500 (trim = 
0°) down to 800 µs (trim = 270°) for closing 
the gate. Trim potentiometer P2 varies the sig
nal  duration  between  1,590  (trim  =  0°)  and 
2,100 µs (trim = 270°). 

When opening or closing the gate the signal du
ration  changes:  when  closing  it  gets  shorter, 
when  opening  it  gets  longer.  In  both  cases 
movement stops when the lower and upper limit 
is reached.

The speed of the in- 
and  decreases  is 
determined  by  the 
trim  potentiometer 
P3. On P3 = 0° in
crease  and  de
crease lasts for sev
en seconds. If P3 = 
270°  the  whole 
movement  lasts 
only  one  second. 
Changes  of  P3 
come  only  into  ef
fect  when the gate 
is moving.

1.4 Variations of the hardware

If you need two LEDs (one for each side of the gates): just connect the second LED with 
its own current-limiting resistor to PB0 and PB1. Up to seven LEDs can be driven by these 
two outputs pins.

If you need only two, you can also connect them serially and reduce the resistor down to 
100 Ω (with 5 V operating voltage). Or if you need four: connect two times two LEDs seri
ally and with two resistors of 100 Ω to PB0 and PB1. Up to 14 LEDs can be driven in this 
way.

If, instead of two push buttons, you want to use only one switch that is switching to close 
or open the gate, use a one with a middle pin. Connect the middle pin to ground (GND) 
and the two other pins to the INT0 and PCINT7 input pins. That works fine, too.



Other extensions, such as receiving and decoding remote control signals on e.g. the PA3 
input pin, require some additional software. Controller-internal hardware such as the 8 bit 
timer TC0, flash storage space, SRAM for storage or six upper registers, of which two reg
ister pairs can be used as pointers, and six lower registers are still available. Enough re
sources that allow for own extensions.

2 Mounting
Mounting comes in two versions, V2 and V3. The only difference between those concerns 
the pins for the servo motor:

• Version 2 with three 1 mm pins in 5.12 mm distance, with GND and VCC on the left 
and right pins,

• Version 3 with three pins in 2.54 mm distance, with GND on the right pin, VCC in 
the mid and the PWM signal on the left pin.

The latter is for plugging in standard servomotors directly. Both versions are 50-by-40 mm 
in size single-sided.

2.1 PCB Version 2

Two 
bridges 
have  to 
be  sol
dered in 
this ver
sion.

2.2 PCB Version 3

Three  bridges  have  to  be  soldered  in  this 



version.

The final product looks like that:

3 Software
The software comes in two versions, too. Version 1 is the older original source from 2004, 
translated to english.  Version 2 is  more recent, makes better use of the available re
sources in the ATtiny24 and is my favored. The functionality is nearly the same. Both 
source codes are in assembler.

3.1 Source code version 1

This  version  of  the  source  code  can  be  downloaded  here and  can  be  viewed  in  the 
attachment. 

3.1.1 Functioning of the software, version 1

See the extensive further comments in the source code by opening it in an editor for de
tailed descriptions of the functioning. On start up of the controller the current status of the 
trim potentiometer is read in, the 64 values on each channel are summed up (0..65,535) 
and the positions are converted as follows: 

• Trim 1: The sum is divided by 128 and 900 is added. This yields the lower position 
of the gate as pulse duration for the servomotor in micro-seconds. 

• Trim 2: The sum value is divided by 128 and (2100 - 511) is added. This yields the 
open position as pulse duration in micro-seconds. 

• Trim 3: The sum value is divided by 256, multiplied by 24, divided by 256 and one 
is added. With this value the opening and closing value is changed. 

The three values are written to registers.

On start up the internal hardware of the controller is set up: 

• The AD converter is started with a clock prescaler of 128 and the conversion com
plete interrupt enabled. For the three AD port pins the digital drivers are switched 
of to reduce power consumption. 

• The port pins for the up/down push buttons are configured as input pins with the 
pull-up resistors switched on. Both input pins trigger external interrupts (INT0 resp. 
PCINT0). 

• The two port pins for the LED are configured as outputs. 
• The  16  bit  timer  TC1  is  configured  as  CTC  (Clear  Timer  on  Compare)  with  a 

prescaler value of 1. On compare match the TC1COMPA interrupt is executed. The 
output OC1A is toggled on compare match. 

http://www.avr-asm-tutorial.net/avr_en/apps/gate_tn24/gate_tn24_v1.asm


3.1.2 Interrupts

All further execution is interrupt-controlled. 

• The AD converter reads the last result in 10 bit mode and adds the result to a sum 
value in  SRAM.  Following each conversion  the  input  MUX channel  is  increased, 
channel 0 restarts  if  two is  exceeded. If  64 rounds are reached, the sums are 
copied to registers, cleared for restart and a flag is set. 

• The timer interrupt of TC1 first checks if the OC1A pin is high (active signal is start
ed) or low (inactive period). Depending from that, either the duration of the short 
active period or the long inactive period duration is loaded to the compare port reg
ister. Both durations add up to 20,000 µ. If the signal output is low, a flag is set. 

• INT0 clears the upward flag, PCIN0 sets it. If the motor is inactive, the gate move
ment is started. If the toggle bit is cleared, the output OC1A is set to toggle. 

All further actions are triggered via flags. 

3.1.3 Flags

Following wake-up of the controller by interrupts (sleep mode idle) flags are consulted to 
check if handling is required. Following flag handling the controller is send to sleep again.

The following handling routines are triggered: 

• If the ADC flag is set, the conversion of the values is performed. If one of the posi
tion values has changed and if the gate is completely closed or opened the new val
ue is written immediately for ten times to the servomotor (adjust mode). 

• If the timer flag is set and if gate movement is active adding resp. subtracting the 
delta value takes place. If the upper or lower limit is reached the gate movement 
flag is cleared and a post-PWM period starts. If that period is over, the toggle flag is 
cleared and the OC1A output will be cleared on compare match. Depending from 
the phase in which the motor state is, the LED is controlled. 

3.2 New and improved version 2 of the source code

At this location the version V2 of the assembler source code is available, it can be read in 
the attachment.

The new version has the following improved properties:

• The 16 bit timer TC1 now operates in fast PWM mode (mode 14), relieving the 
compare match A interrupt from the task of switching compare A match values be
tween active and inactive signal periods.

• The ICR port register in TC1 is now used as CTC value instead of compare match A. 
ICR is fixed at 20,000 µs, compare match A now works as value selection for the 
active signal duration. Pin control on OC1A (port pin PA6) is selected to set on bot
tom count and to clear on compare match A.

• The PWM pulse width is now between 800 and 2,200 (previous version: 900 and 
2,100). This is sufficient for available servomotors.

• The duration over which the gate opens and closes (between one and seven sec
onds) is now independent from the selected opening and closing positions. This is 
achieved by dividing the range difference between the closed and opening position 
by the third trim potentiometer's position (7 seconds = 350 PWM cycles on trim 3 
in 0° and MSB=0, 1 second = 50 PWM cycles on trim 3 in 270° and MSB=255). 
This allows for a constant closure and opening time instead of speed depending 
from the positions. Note that in the highest speed selection (trim 3 in 270° posi
tion) this only works correct for opening angles of 31° upwards due to rounding ef
fects.

http://www.avr-asm-tutorial.net/avr_en/apps/gate_tn24/gate_tn24_v2.asm


• Up- or down-movement of the motor is now controlled via the T-flag in the status 
register port. This decision simplifies external interrupt processing to react on push 
button events. Flags now are generally simplified.

• ADC measurements are now organized 64 measurements for each channel in a 
row, which reliefs the ADCRDY interrupt service routine from the task of switching 
between channels. Channel selection now is done outside the interrupt service rou
tine. Because only one sum and not three different sums have to be calculated, 
summing in SRAM is avoided, also an optimization.

The new version is much simpler in its structure, much easier to understand and consider
ably shorter than the previous version.

This is the PWM signal 
generated  on  the 
OC1A port pin. This is 
rather exact.

4. Pictures and videos
Here is a picture after the servomotor reached its closed position and the red LED is per
manently on.



And here the gate is in its open position and the green LED is on.

In this video the gate moves slow from the open to the closed and back to the open posi
tion. In this video the movement is fast. If those videos do not open automatically in your 
browser, download them and start them locally in your favored video player.

©2014-2018 by http://www.avr-asm-tutorial.net 

http://www.avr-asm-tutorial.net/avr_en/apps/gate_tn24/gate_tn24_slow.MOV
http://www.avr-asm-tutorial.net/avr_en/apps/gate_tn24/gate_tn24_fast.MOV
http://www.avr-asm-tutorial.net/


Attachment: Source code V1

Software for the model railroad 
crossing gate with an ATtiny24 and a 
servermotor (version 1.0)
The source code of the software for the model railroad crossing is available here, an im
proved version is . 

;
; ***************************************************
; * Railway gate with an ATtiny24 and a servo motor *
; * V1, 19.07.2014, translated 17.03.2018           *
; * (C)2014-2018 by http://www.avr-asm-tutorial.net *
; ***************************************************
;
; Include file for the AVR type
.NOLIST
.INCLUDE "tn24def.inc" ; Header file for ATtiny24
.LIST
;
; Debug switches (final: all zero)
.equ debugparam=0
;
; ============================================
;   H A R D W A R E   I N F O R M A T I O N 
; ============================================
;
;             ______________
;            /   ATtiny24   |
;      +U o--|VCC        GND|--o -U
; Duo LED    |              |
;   Red   o--|PB0       ADC0|--o Lower trim
;   Anode    |              |
;   Red   o--|PB1       ADC1|--o Upper trim
;   Cat      |              |
;   RESET o--|RESET     ADC3|--o Speed trim
;            |              |
;     DWN o--|INT0       PA4|--o NC
;            |              |
;      UP o--|PCINT7    USC0|--o SCK
;            |              |
; PWM-SIG o--|OC1A/MOSI MISO|--o MISO
;  & MOSI    |______________|
;
;
; ============================================
;      P O R T S   A N D   P I N S 
; ============================================
;
; Ports for LED control
.EQU pLedO = PORTB
.EQU pLedI = PINB
.EQU pLedD = DDRB

file:///C:/Users/gerd/Documents/9_websites/gsc-da/html/avr-asm/avr_en/apps/gate_tn24/gate_tn24_v1.asm


.EQU bLedPlO = PORTB0

.EQU bLedPlD = DDB0

.EQU bLedPlI = PINB0

.EQU bLedMiO = PORTB1

.EQU bLedMiD = DDB1

.EQU bLedMiI = PINB1
; Ports fuer Tastensteuerung
.EQU pDwnO = PORTB
.EQU pDwnD = DDRB
.EQU bDwn = PORTB2
.EQU pUpO = PORTA
.EQU pUpD = DDRA
.EQU pUpI = PINA
.EQU bUpO = PORTA7
.EQU bUpD = DDA7
.EQU bUpI = PINA7
; Ports fuer PWM-Signal
.EQU pPwmO = PORTA
.EQU pPwmI = PINA
.EQU pPwmD = DDRA
.EQU bPwmO = PORTA6
.EQU bPwmD = DDA6
.EQU bPwmI = PINA6
;
; =============================================
;    C O N S T A N T S   T O   A D J U S T
; =============================================
;
.EQU cMinMin = 900 ; us shortest PWM signal duration
.EQU cMaxMax = 2100 ; us longest PWM signal duration
.EQU cMinMed = 1000 ; us lower regulation limit
.EQU cMaxMed = 2000 ; us upper regulation limit
.EQU cMid = (cMaxMax+cMinMin) / 2 ; Middle position
.EQU cPwm = 20000 ; us total PWM duration
.EQU cSpeedHigh = 1000000 ; us duration fastest closing
.EQU cSpeedLow = 7000000 ; us duration slowest closing
;
; ===============================================
;  F I X E D + D E R I V E D   C O N S T A N T S 
; ===============================================
;
.EQU cMaxMin = cMaxMax - 511 ; Lower limit closing
.EQU cRoundsFast = cSpeedHigh / cPwm
.EQU cRoundsSlow = cSpeedLow / cPwm
.EQU cAdderFast = (cMaxMed - cMinMed) / cRoundsFast
.EQU cAdderSlow = 1 ; Slow adder
.EQU cSpeedMulti = cAdderFast - cAdderSlow
;
; ===================================
;  P R O G R A M   S T R U C T U R E
; ===================================
;
; All timing is based on a clock frequency of 1 MHz,
; which is the default clock frequency with CLKDIV8
; set and the 8 MHz RC oscillator selected.
;
; ADC values
; Measuring the trim potentiometer settings runs with
; an ADC clock prescaler of 128. 64 measurements per
; channel are added up in the registers R0 to R5 and
; averaged.
; From this a measuring frequency of 1 Mhz / 128 /
; 13 clock cycles per conversion / 3 channels / 64
; single measurements = approx. 3 Hz results.
; Calculation of those measurements goes as follows:
; Trim 1: Sum/128+cMinMin(900) ==> rClose (Minimum)
; Trim 2: cMaxMax(2100)-Sum/128 ==> rOpen (Maximum)
; Trim 3: (Sum/256*cSpeedMulti)/256+1 ==> rDelta
;   (Speed), values vary between 1 and 23
;   (1 = 20 seconds closure time, 23 = one second
;   closure time
;
; Timer1: Generation of the PWM signal
;   TC1 is clocked by the clock signal and a prescaler
;   of 1, one tick is therefore 1 us.
;   TC1 runs as CTC and generates the PWM signal on the
;   output pin OC1A on PA6. The output pin toggles on
;   compare match A, TC1 restarts and triggers a
;   compare match A intterupt. Depending from the state
;   of the output pin either the duration in rCtcAct or



;   in rCtcIna is written to the compare register A. The
;   resulting looks as follows:
;
;       _______                   _______                   ______
; _____|       |_________________|       |_________________|
;       rCtcAct    rCtcIna        rCtcAct      rCtcIna
;      |<======= 20.000 us =====>|<======= 20.000 us =====>|
;
; rCtcAct = 900 .. 2.100 us
;   Lower limit adjustable with trim 1 between 900 and 1410 us
;   Upper limit adjustable with trim 2 between 1590 and 2100 us
; rCtcIna = 20.000 - rCtcAct
;
; rCtcAct plus rCtcIna always yields 20.000 us, a 50 Hz signal.
; When the OC1A pin is set (after writing rCtcIna) the bPwm flag
; is set, which triggers execution of the routine PwmRdy. This
; determines if the gate is currently moving or not.
; If this is not the case, 25 following PWM signal bursts are
; sent with the current position (0.5 seconds). After that the
; PWM signal output is cleared (by clearing the output pin on
; compare match).
; If the gate is moving the duration of the PWM signal is increased
; by the content of rDelta, if the gate opens, or decreased if
; the gate closes. If, on opening, the upper limit in rOpen is
; exceeded the LED is switched to permanently green and the
; post-PWM period is initiated.
; If, on closing, the lower limit in rClose is reached the LED
; is switched to permanent red and the post-PWM period initiated.
; Bei aktiver Schranke wird die Dauer des PWM-Signals um den Betrag
; rDelta verlaengert (Schranke oeffnet sich) bzw. verkuerzt (Schranke
; schliesst sich). Ist beim Oeffnen die obere Grenze in rOben ueber-
; schritten wird die LED auf gruenes Dauerlicht geschaltet und der
; Nachlauf eingeleitet. Ist beim Schliessen die untere Grenze unter-
; schritten wird die LED auf rotes Dauerlicht geschaltet und der
; Nachlauf eingeleitet.
; If neither the upper nor the lower limit has been reached the
; LED blinks with 0.2 Hz.
;
; Triggering opening and closing of the gate
; If push button 1 is activated, an INT0 interrupt is triggered.
; This clears the upward flag.
; If push button 2 is activated, the PCINT0 interrupt is triggered.
; Following both interrupts and if the toggle bit of TC1 is not
; set, the rCtcIna value is written to the compare A port, the
; toggle mode in TC1 is set and the compare match A interrupt is
; enabled. Finally the "Gate active" flag is set.
;
; Adjusting
; If the value of the up- or down-trim-poentiometer is changed,
; the new value is immediately applied. If the gate is inactive
; (bActive flag cleared), the gate only moves if it is completely
; open (last movement upwards) or completely closed (last movement
; downwards).
; Changes of the speed trim potentiometer come only into effect
; if the gate moves.
;
; Start conditions
; On start up the gate is moved to to its middle position and then
; upwards to the upper limit.
;
; ============================================
;   R E G I S T E R   D E F I N I T I O N S
; ============================================
;
; R0 .. R5 : ADC sum values, three channels
.DEF rCtcActL = R6 ; Active signal duration, us
.DEF rCtcActH = R7
.DEF rCtcInaL = R8 ; Inactive signal duration, us
.DEF rCtcInaH = R9
.DEF rCloseL = R10 ; Lower limit PWM signal
.DEF rCloseH = R11
.DEF rOpenL = R12 ; Upper limit PWM signal
.DEF rOpenH = R13
.DEF rDelta = R14 ; Step length for gate movement
.DEF rSreg = R15 ; for SREG interim storage in ints
.DEF rmp = R16 ; Multi purpose register
.DEF rimp = R17 ; Multi purpose inside interrupts
.DEF rimp2 = R18 ; 16 bit adder register in interrupts
.DEF rAdcCnt = R19 ; Counter for ADC measurements
.DEF rFlg = R20 ; Flag register



        .equ bActive = 0 ; Gate is moving
        .equ bHigh = 1 ; Gate is moving up
        .equ bAdc = 2 ; ADC sequence complete
        .equ bPwmIna = 3 ; Inactive PWM signal, handle flag
        .equ bOpen = 4 ; Gate is open
        .equ bClose = 5 ; Gate is closed
        .equ bToggle = 6 ; CTC toggle is active
.DEF rPwmCtr = R21 ; Downcounter for post-PWM phase
; R22 .. R25 not used
; X = XH:XL Multiplication for speed adjustment
; Y = YH:YL ADC pointer to SRAM
; Z = ZH:ZL Multi purpose register pair outside interrupts
;
; ============================================
;       S R A M   D E F I N I T I O N S
; ============================================
;
.DSEG
.ORG  0X0060
sAdc: .Byte 6
;
; ==============================================
;   R E S E T   A N D   I N T   V E C T O R E N
; ==============================================
;
.CSEG
.ORG $0000
        rjmp Main ; Reset-Vector
        rjmp IntDwn ; Int Vector INT0
        rjmp IntUp ; Int Vector PCINT0
        reti ; Int Vector PCINT1
        reti ; Int Vector WDT
        reti ; Int Vector TC1 Capture
        rjmp IntPwmA ; Int Vector TC1 Compare A
        reti ; Int Vector TC1 Compare B
        reti ; Int Vector TC1 OVF
        reti ; Int Vector TC0 Compare A
        reti ; Int Vector TC0 Compare B
        reti ; Int Vector TC0 OVF
        reti ; Int Vector Ana_Comp
        rjmp IntAdc ; Int Vector ADC
        reti ; Int Vector EEPROM Ready
        reti ; Int Vector USI START
        reti ; Int Vector USI Overflow
;
; ==========================================
;    I N T E R R U P T   S E R V I C E
; ==========================================
;
; Downwards push button
IntDwn:
        in rSreg,SREG ; Save SREG
        cbr rFlg,1<<bHigh ; Clear upward flag
        rjmp IntUD
;
; Upwards push button
IntUp:
        in rSreg,SREG ; Save SREG
        sbic pUpI,bUpI ; Falling edge?
        reti ; No, ignore
        sbr rFlg,1<<bHigh ; Set upwards flag
IntUD:
        ; Start move cycle
        sbrc rFlg,bActive ; Already active?
        rjmp IntUD1 ; Yes, ignore
        sbr rFlg,1<<bActive ; Set gate aktice flag
        sbrc rFlg,bToggle ; Skip if not in toggle mode
        rjmp IntUD1 ; Toggle-Mode is on
        ldi rPwmCtr,1 ; Start blink counter
        out OCR1AH,rCtcInaH ; Compare match inactive duration
        out OCR1AL,rCtcInaL
        ldi rimp,1<<COM1A0 ; Enable toggle of output pin
        out TCCR1A,rimp
        ldi rimp,1<<OCIE1A ; Enable TC1COMPA interrupt
        out TIMSK1,rimp
        sbr rFlg,1<<bToggle
IntUD1:
        out SREG,rSreg ; Restore SREG
        reti
;



; TC1 Compare match A interrupt service routine
IntPwmA:
        in rSreg,SREG ; Save SREG
        sbic pPwmI,bPwmI ; Skip if PWM signal=0
        rjmp IntPwmA1
        out OCR1AH,rCtcInaH ; Long inactive pause
        out OCR1AL,rCtcInaL
        sbr rFlg,1<<bPwmIna ; Set flag
        rjmp IntPwmA2
IntPwmA1:
        out OCR1AH,rCtcActH ; Short active signal
        out OCR1AL,rCtcActL
IntPwmA2: 
        out SREG,rSreg ; Restore SREG
        reti
;
; AD conversion complete interrupt service routine
IntAdc:
        in rSreg,SREG ; Save SREG
        in rimp,ADCL ; Read LSB result
        ld rimp2,Y ; Read current sum LSB
        add rimp,rimp2 ; Add LSB to current
        st Y+,rimp ; and store in SRAM
        in rimp,ADCH ; Read MSB result
        ld rimp2,Y ; Read MSB current sum
        adc rimp,rimp2 ; Add MSBs with carry
        st Y+,rimp ; Store MSB
        cpi YL,6 ; End of round?
        brcs IntAdcRet ; No, continue
        clr YL ; Restart
        dec rAdcCnt ; Count rounds down
        brne IntAdcRet
AdcCopy:
        ; End of ADC measuring cycle
        ldi YL,LOW(sAdc) ; Copy to SRAM
        st Y+,R0
        st Y+,R1
        st Y+,R2
        st Y+,R3
        st Y+,R4
        st Y,R5
        clr YL
        clr R0 ; Clear sums 
        clr R1
        clr R2
        clr R3
        clr R4
        clr R5
        sbr rFlg,1<<bAdc ; Set flag
IntAdcRet:
        mov rimp,YL ; Copy MUX channel
        lsr rimp ; Divide by two
        out ADMUX,rimp ; MUX to next channel
        ; Start next conversion
        ldi rimp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
        out ADCSRA,rimp
        out SREG,rSreg ; Restore SREG
        reti
;
; ============================================
;    M A I N   P R O G R A M   I N I T
; ============================================
;
Main:
; Init stack
        ldi rmp, LOW(RAMEND) ; Init LSB stack
        out SPL,rmp
; Init port bits
        ; Init LED pins
        sbi pLedD,bLedPlD ; LED pins as output
        sbi pLedD,bLedMiD
        sbi pLedO,bLedPlO ; Turn red LED on
        cbi pLedO,bLedMiO
        ; Init the push button pins
        cbi pDwnD,bDwn ; Close push button as input
        sbi pDwnO,bDwn ; Pull up resistor on
        cbi pUpD,bUpD ; Close push button as input
        sbi pUpO,bUpO ; Pull up resistor on
        ; Init PWM output pin
        sbi pPwmD,bPwmD ; PWM output pin as output



        cbi pPwmO,bPwmO ; To low
; Init ADC
        ; Single measurement at start up
.if debugparam==1
        rjmp debugcalc
        nop
        .endif 
        rcall AdcGet ; Get first cycle
        ldi rmp,HIGH(cMid) ; Position to middle
        mov rCtcActH,rmp
        ldi rmp,LOW(cMid)
        mov rCtcActL,rmp
        clr rFlg ; Clear flags
        sbr rFlg,(1<<bActive)|(1<<bHigh) ; Move gate up
        ldi rPwmCtr,1 ; Init LED blink counter
        rcall PwmRdyO ; Blink and correct Inactive duration
        ; Prepare ADC sequence
        ldi YH,HIGH(sAdc) ; Pointer to SRAM
        ldi YL,LOW(sAdc)
        ldi rmp,0 ; MUX to channel 0
        out ADMUX,rmp
        ldi rmp,(1<<ADC0D)|(1<<ADC1D)|(1<<ADC0D) ; Digital input driver disable
        out DIDR0,rmp
        clr rmp ; Results not left adjust
        out ADCSRB,rmp
        ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
        out ADCSRA,rmp
; Init timer 1
        out OCR1AH,rCtcInaH ; CTC to inactive duration
        out OCR1AL,rCtcInaL
        ldi rmp,1<<COM1A0 ; Toggle output A on compare match
        out TCCR1A,rmp
        ldi rmp,(1<<WGM12)|(1<<CS10) ; CTC mode, prescaler = 1
        out TCCR1B,rmp
        ldi rmp,1<<OCIE1A ; TC1 Compare Match A Interrupt Enable
        out TIMSK1,rmp
        sbr rFlg,1<<bToggle ; Toggle is active
; Init PCINT7
        ldi rmp,1<<PCINT7 ; Pin-Change Int push button 2
        out PCMSK0,rmp
; Init external interrupts and Sleep mode idle
        ldi rmp,(1<<SE)|(1<<ISC01) ; Enable sleep and INT0
        out MCUCR,rmp
        ; External interrupts
        ldi rmp,(1<<INT0)|(1<<PCIE0) ; Enable INT0 and PCINT0 interrupts
        out GIMSK,rmp
        sei ; Enable interrupts
;
; ============================================
;        P R O G R A M - L O O P
; ============================================
;
Loop:
        sleep ; Send to sleep
        nop ; Dummy for wake-up
        sbrc rFlg,bPwmIna ; PWM flag set?
        rcall PwmRdy ; Yes, handle PWM
        sbrc rFlg,bAdc ; ADC flag set?
        rcall AdcRdy ; Yes, handle ADC
        rjmp loop ; Back to loop
;
; PWM ready
PwmRdy:
        cbr rFlg,1<<bPwmIna ; Clear flag
        sbrc rFlg,bActive ; Skip if gate inactive
        rjmp PwmRdyA ; Gate active
        ; Gate inactive
        sbrs rFlg,bToggle ; Skip if toggle is on
        ret ; Toggle is off, return
        ; TC1 toggle is on
        dec rPwmCtr ; Count post-PWM down
        brne PwmRdy1 ; Not finished yet
        ldi rmp,1<<COM1A1 ; Set TC output pin to clear
        out TCCR1A,rmp ; Switch toggle off
        clr rmp ; Disable TC1 interrupt
        out TIMSK1,rmp
        cbr rFlg,1<<bToggle ; Toggle flag off
PwmRdy1:
        ret
PwmRdyA: ; Gate ist active



        sbrc rFlg,bHigh ; Skip if gate is closed
        rjmp PwmRdyH ; Gate upwards
        ; Gate downwards
        sub rCtcActL,rDelta ; Reduce PWM signal duration
        brcc PwmRdyA1 ; No carry
        dec rCtcActH ; MSB downwards
PwmRdyA1:
        cp rCtcActL,rCloseL ; Compare with lower limit
        cpc rCtcActH,rCloseH
        brcc PwmRdyA2 ; Not yet reached
        mov rCtcActL,rCloseL ; Reached, set close value
        mov rCtcActH,rCloseH
        cbr rFlg,1<<bActive ; End gate movement
        sbr rFlg,1<<bClose ; Set close flag
        sbi pLedO,bLedPlO ; Red LED permanently on
        cbi pLedO,bLedMiO
        ldi rPwmCtr,25 ; Start repeat counter for post-PWM
        rjmp PwmRdyK ; Correct duration PWM
        ; Not yet on close position
PwmRdyA2:
        cbr rFlg,1<<bClose ; Close flag off
        rjmp PwmRdyO ; Blink and correct PWM duration
PwmRdyH:
        ; Gate upwards
        add rCtcActL,rDelta ; Add delta to signal duration
        brcc PwmRdyH1 ; No carry
        inc rCtcActH ; Carry to MSB
PwmRdyH1:
        cp rCtcActL,rOpenL ; Compare with upper limit
        cpc rCtcActH,rOpenH
        brcs PwmRdyH2 ; Not yet reached
        mov rCtcActL,rOpenL ; Reached, set upper limit
        mov rCtcActH,rOpenH
        cbr rFlg,1<<bActive ; End gate movement
        sbr rFlg,1<<bOpen ; Gate open flag
        cbi pLedO,bLedPlO ; LED permanently green
        sbi pLedO,bLedMiO
        ldi rPwmCtr,25 ; Start repeat counter for post-PWM
        rjmp PwmRdyK ; Correct duration of PWM inactive
PwmRdyH2:
        cbr rFlg,1<<bOpen ; Open flag off
        rjmp PwmRdyO ; LED blink and CTC-Inactive correction 
PwmRdyO:
        ; LED blink
        cbi pLedO,bLedMiO
        dec rPwmCtr ; Counter for blink LED
        brne PwmRdyK ; Correct inactive duration PWM
        ldi rPwmCtr,12 ; Restart counter
        sbic pLedI,bLedPlI ; Skip if LED is off
        rjmp PwmRdyO1 ; LED is on
        sbi pLedO,bLedPlO ; LED on
        rjmp PwmRdyK
PwmRdyO1:
        cbi pLedO,bLedPlO ; LED off
PwmRdyK:
        ; Correct inactive duration PWM
        ldi rmp,LOW(cPwm) ; Total duration - Active duration
        sub rmp,rCtcActL
        mov rCtcInaL,rmp ; Copy to inactive duration
        ldi rmp,HIGH(cPwm)
        sbc rmp,rCtcActH
        mov rCtcInaH,rmp
        ret
;
; ADC measuring sequence ended
;
AdcRdy:
        cbr rFlg,1<<bAdc
        push rCloseL
        push rCloseH
        push rOpenL
        push rOpenH
        rcall CalcSpeed
        pop XH
        pop XL
        pop ZH
        pop ZL
        sbrc rFlg,bActive ; Skip if gate still active
        ret ; Yes
        sbrc rFlg,bToggle ; Skip if toggle active



        ret ; Yes
        ldi rmp,LOW(cMid) ; Below or above middle?
        cp rCtcActL,rmp
        ldi rmp,HIGH(cMid)
        cpc rCtcActH,rmp
        brcc AdcRdyU 
        ; Gate is below middle
        mov rCtcActH,rCloseH
        mov rCtcActL,rCloseL
        rjmp AdcRdyS ; Start move cycle
AdcRdyU:
        ; Gate is above middle
        mov rCtcActH,rOpenH
        mov rCtcActL,rOpenL
AdcRdyS:
        ; Start adjustment
        rcall PwmRdyK ; Calculate inactive duration
        ldi rPwmCtr,10 ; Start post-PWM cycle
        ldi rmp,1<<COM1A0 ; Start toggle
        out TCCR1A,rmp
        ldi rmp,1<<OCIE1A ; Start interrupt
        out TIMSK1,rmp
        sbr rFlg,1<<bToggle ; Set toggle flank
        ret
;
; =============================================
;     A S Y N C H   S U B R O U T I N E S
; =============================================
;
; Get ADC results during init
AdcGet:
        ldi rmp,(1<<ADC0D)|(1<<ADC1D)|(1<<ADC2D) ; Digital drivers off
        out DIDR0,rmp
        clr rmp ; Result not left adjusted
        out ADCSRB,rmp
        ldi YH,HIGH(sAdc) ; Channel counter Y points to SRAM
        ldi YL,LOW(sAdc)
GetAdc1:
        clr R0 ; Clear result
        clr R1
        ldi rAdcCnt,64 ; 64 measurements
        mov rmp,YL ; MUX channel
        lsr rmp
        andi rmp,0x03 ; Isolate channel bits 
        out ADMUX,rmp
GetAdc2:
        ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
        out ADCSRA,rmp
GetAdc3:
        sbic ADCSRA,ADSC ; Wait for result
        rjmp GetAdc3
        in rmp,ADCL ; Add LSB to result
        add R0,rmp
        in rmp,ADCH ; Add MSB to result
        adc R1,rmp
        dec rAdcCnt
        brne GetAdc2 ; Again convert
        st Y+,R0 ; Store LSB
        st Y+,R1 ; Store MSB
        cpi YL,sAdc+6 ; All three channels read?
        brcs GetAdc1
;
; Calculate delta from speed position
CalcSpeed:
        lds rmp,sAdc+5 ; ADC result MSB channel 3
        clr XH ; Multiplicator MSB
        ldi XL,cSpeedMulti ; LSB = Multiplicator for speed
        clr ZH ; Z is result
        clr ZL
CalcSpeed1:
        lsr rmp ; Bit = 1?
        brcc CalcSpeed2 ; No, do not add
        add ZL,XL ; Add LSB multiplicator
        adc ZH,XH ; Add MSB and carry
CalcSpeed2:
        lsl XL ; Double multiplicator
        rol XH
        tst rmp ; Already done?
        brne CalcSpeed1 ; No, further bits
        ldi rmp,cAdderSlow



        add rmp,ZH ; Add MSB of the result
        mov rDelta,rmp
;
; Lower and upper trim potentiometer
AdcConv:
        lds rmp,sAdc ; Read LSB lower trim
        lds ZL,sAdc+1 ; Read MSB upper trim
        clr ZH
        lsl rmp ; Upper bit LSB into lower bit LSB
        rol ZL ; Rotate into LSB
        rol ZH ; And into MSB
        ldi rmp,LOW(cMinMin) ; Minimum duration lower limit
        add ZL,rmp ; Add to result
        ldi rmp,HIGH(cMinMin)
        adc ZH,rmp
        andi ZL,0xFC ; Clear lowest two bits
        mov rCloseH,ZH ; Copy to closure position
        mov rCloseL,ZL
        lds rmp,sAdc+2 ; Read LSB upper trim
        lds ZL,sAdc+3 ; Read MSB upper trim
        clr ZH
        lsl rmp ; Upper bit LSB to lowest bit LSB
        rol ZL ; Rotate bit into LSB in ZL
        rol ZH ; And into MSB
        ldi rmp,LOW(cMaxMin) ; LSB maximum duration upper position
        add ZL,rmp ; Add to LSB
        ldi rmp,HIGH(cMaxMin)
        adc ZH,rmp ; Add MSB and carry
        andi ZL,0xFC ; Clear lowest two bits
        mov rOpenL,ZL
        mov rOpenH,ZH
        ret
.if debugparam==1
debugcalc:
        ldi rmp,0xFF
        ldi ZH,HIGH(sAdc)
        ldi ZL,LOW(sAdc)
        st Z+,rmp
        st Z+,rmp
        st Z+,rmp
        st Z+,rmp
        st Z+,rmp
        st Z+,rmp
        rjmp CalcSpeed
        .endif
;
; End of source code
; Copyright
.db "(C)2014 by Gerhard Schmidt  " ; human readable
.db "C(2)10 4ybG reahdrS hcimtd  " ; wordwise
;

Attachment: Source code V2

Applications of
AVR Single chip 

controllers AT90S, 
ATtiny, ATmega and 

ATxmega

Servomotor 
controller with an 

ATtiny24

Software for the model railroad 
crossing gate with an ATtiny24 and a 
servermotor, version 2
This source code of the software for the model railroad crossing is available here. 

;
; ***********************************
; * Gate controller with ATtiny24   *
; * (C)2018 by avr-asm-tutorial.net *
; ***********************************
;
.nolist
.include "tn24adef.inc" ; Define device ATtiny24A
.list
;
; **********************************
;    D E B U G   S W I T C H E S
; **********************************
;
; All switches have to be 0 in the final version
;
; Test the ADC conversion routines
.equ DebugAdc = 0 ; Debug the ADC routines
  .equ cAdcChannel = 1 ; Channel (0..2)
  .equ cAdcResult = 65535 ; Result of the ADC sum, 0 to 6535
  .equ cOpen = 1000 ; Open position, 800 to 1,500
  .equ cClose = 2000 ; Close position, 1,500 to 2,200
.equ DebugPosition = 0 ; Follow the current position
  .equ DebugOpen = 0 ; Choose this to follow the open position
  .equ DebugClose = 0 ; Choose this to follow the close position
.equ DebugOpenClose = 0 ; Open and close the gate
;
; Check errors
.if (DebugPosition == 1) && (DebugOpen == 0) && (DebugClose == 0)
  .error "No positioning selected!"
  .endif
.if (DebugPosition == 1) && (DebugOpenClose == 1)
  .error "Positioning and Open/Close settings conflicting!"
  .endif
;
; **********************************
;        H A R D W A R E
; **********************************
;
; Device: ATtiny24A, Package: 14-pin-PDIP_SOIC
;
;            _________
;         1 /         |14
; + 5 V o--|VCC    GND|--o 0 V
; DuoLED  2|          |13
; An rd o--|PB0   ADC0|--o Trim 1
;         3|          |12

; An gn o--|PB1   ADC1|--o Trim 2
;         4|          |11
; RESET o--|RESET ADC2|--o Trim 3
;         5|          |10
; BtnUp o--|INT0   PA3|--o NC
;         6|          |9
; BtnDwno--|PCINT7 SCK|--o ISP6-SCK
;         7|          |8
;   PWM o--|OC1A  MOSI|--o ISP6-MOSI
; ISP6-MISO|__________|
;
;
; **********************************
;  P O R T S   A N D   P I N S
; **********************************
; Duo LED control
.equ pLedO = PORTB ; LED output port
.equ pLedD = DDRB  ; LED direction port
.equ pLedI = PINB  ; LED input port
.equ bLedAnRdO = PORTB0 ; Red anode output pin
.equ bLedAnRdD = DDB0 ; Red anode direction pin
.equ bLedAnGnO = PORTB1 ; Green anode output pin
.equ bLedAnGnD = DDB1 ; Green anode direction pin
;
; Push button down control
.equ pBtnDO = PORTB ; Push button up output port
.equ pBtnDD = DDRB ; Push button up direction port
.equ pBtnDI = PINB ; Push button up input port
.equ bBtnDO = PORTB2 ; Push button up output pin
.equ bBtnDD = DDB2 ; Push button up direction pin
.equ bBtnDI = PINB2 ; Push button up input pin
;
; Push button up control
.equ pBtnUO = PORTA ; Push button down output port
.equ pBtnUD = DDRA ; Push button down direction port
.equ pBtnUI = PINA ; Push button down input port
.equ bBtnUO = PORTA7 ; Push button down output pin
.equ bBtnUD = DDA7 ; Push button down direction pin
.equ bBtnUI = PINA7 ; Push button down input pin
;
; PWM signal
.equ pPwmO = PORTA ; PWM output port
.equ pPwmD = DDRA ; PWM direction port
.equ bPwmO = PORTA6 ; PWM output pin
.equ bPwmD = DDA6 ; PWM direction pin
;
; **********************************
;   A D J U S T A B L E   C O N S T
; **********************************
;
; PWM signal basic properties
.equ cPwmMin = 1000 ; Mimimum duration, ms
.equ cPwmMax = 2500 ; Maximum duration, ms
.equ cFpwm = 50 ; Frequency of the PWM signal
.equ cPost = 25 ; Post period duration, 0.5 sec
; Up/down movement speed
.equ cSpeedHigh = 1 ; Maximum speed, one second
.equ cSpeedLow = 7 ; Minimum speed, seven seconds
;
; **********************************
;  F I X  &  D E R I V.  C O N S T
; **********************************
;
; Controller clock = Default
.equ clock=1000000 ; Define clock frequency
;
; PWM signal duration, default = 20,000 us
.equ cPwm = clock / cFpwm - 1 ; in us
;
; PWM signal properties derived
; Middle position of the servomotor
; Default: 1,500
.equ cPwmMid = (cPwmMax + cPwmMin + 1) / 2
; Max. +/- for up and down movement
; Default: +/- 500
.equ cPwmDelta = (cPwmMax - cPwmMin +1) / 2
; Default values for startup
.equ cPwmCloseDef = cPwmMid - 50 ; slightly down
.equ cPwmOpenDef = cPwmMid + 50 ; slightly up
.equ cPwmSpeedDef = 1 ; Default speed at startup



;
; Speed calculation from trim 3
.equ cCycleHigh = cFpwm*cSpeedHigh ; Default: 50
.equ cCycleLow = cFpwm*cSpeedLow ; Default: 350
.equ cSpdMul = cCycleLow - cCycleHigh ; Multiplicator
.equ cSpdAdd = cCycleHigh ; Adder to result
;
; *********************************************
;  H O W   T H E   S O F T W A R E   W O R K S
; *********************************************
;
; Relevant registers:
;   rClose: 16 bit register that holds the PWM
;     duration of the motor in closed position,
;     can be between 1,500 and 800.
;   rOpen: 16 bit register that holds the PWM
;     duration of the motor in opened position,
;     can be between 1,500 and 2,200.
;   rDelta: 8 bit register by which the position
;     value is increased (closing selected) or
;     decreased (opening selected), values are
;     between 1 (slowest) and 28 (fastest).
;
; PWM signal and motor position
;   The short PWM period (800 us) moves the servo
;   to the rightmost position (gate fully closed).
;   The long PWM period (2,200 us) moves the servo
;   to the leftmost position (gate fully opened).
;   Trim 1 increases the closed position from the
;   middle position (1,500 us).
;   The trim potentiometer P1 (closure) varies the
;   PWM period between 1,500 (trim 1 at 0 degrees)
;   and 800 us (trim 1 at 270 degrees). The closure
;   value is in rClose.
;   The trim potentiometer P2 (opening) varies the
;   PWM period between 1,500 (trim 2 at 0 degrees)
;   and 2,200 us (trim 2 at 270 degrees). The open
;   value is in rOpen.    
;
; PWM signal generation:
;   16 bit timer TC1 operates as fast PWM generator
;   in mode #14. It runs with a prescaler of 1
;   (at 1 us per tick). The ICR1 register is used
;   to clear the TC1 and to restart the PWM cycle.
;   Those two port registers ICR1H:ICR1L are set
;   to 19,999 to achieve a 50 Hz PWM frequency as
;   required by the servomotor.
;   Output pin OC1A (on Pin PA6) is set on PWM
;   cycle start and cleared on compare match A (non-
;   inverted PWM mode). Compare match A is set to
;   the current motor position in us. Its value
;   comes only into effect when the end of the PWM
;   cycle is reached (TC1 count = 20.000).
;   Also at compare match A the next position of
;   the motor is calculated (see next para).
;
; TC1 compare match interrupt:
;   Moving the gate downwards:
;   If the direction of the gate is currently
;   downwards (T flag in SREG is cleared), the
;   actual compare match A value is compared
;   with the rClose register pair. If both are
;   equal, the post-period part is executed
;   (see below).
;   If the position in the compare match value
;   A differs from the rClose register pair
;   the current position in compare match A
;   is reduced by the register rDelta. If the
;   result is smaller than the rClose value, the
;   rClose value is written to the compare match
;   A port registers. If not, the result of the
;   subtraction is written as new compare match
;   A value. The update of the compare match A
;   value occurs automatically on the end of
;   the current PWM cycle.
;   The post-period counter is set to its
;   initial value cPost. The LED is set to
;   blink in red color.
;
;   Moving the gate upwards:



;   If the direction of the gate is currently
;   upwards (T flag in SREG is high), the actual
;   compare match A value is compared with the
;   value in the rOpen register pair. If both
;   are equal, post-period execution follows
;   (see below).
;   If the compare match value A differs from
;   rOpen, the rDelta register content is
;   added to the compare match A value. If the
;   sum exceeds or equals the rOpen value, the
;   rOpen value is written to the compare
;   match A port registers. If not, the sum is
;   written to the compare match A port
;   registers. The altered compare match A value
;   is coming into effect when the current PWM
;   cycle ends.
;   The post-period counter is set to its
;   initial value cPost. The LED is set to
;   blink in red color.
;
;   Post-period execution (compare match equals
;   rOpen/rClose):
;   If either the compare match A value equals
;   rOpen in downward direction (T flag is clear)
;   or if it equals rClose in downwards direction
;   (T flag is clear) the register content of
;   rPost is checked. If that is already at
;   zero, nothing else happens.
;   If not, it is counted down. If it then
;   reaches zero, the LED is set permanently
;   to green (upwards, T flag set, gate fully
;   open), otherwise permanently to red (down-
;   wards, T flag cleared, gate fully closed).
;
; Measuring and converting the trim potentiometers:
;   Close position trim (P1):
;   64 measurements of ADC channel 0 are summed
;   up. The MSB of the result is multiplied by
;   cPwmDelta (default: 500), which is a 16-by-8
;   bit multiplication). Byte 3 and byte 2 of
;   the result are subtracted from cPwmMid (by
;   default 1,500) and the result is copied to
;   rClose.
;
;   Open position trim (P2):
;   Channel 1 of the ADC is measured 64 times and
;   summed up, the MSB of the result is multi-
;   plied with cPwmDelta (700). Byte 3 and byte
;   2 of the result are added to cPwmMid (1,500)
;   and the result is written to the register
;   pair rOpen.
;
;   Speed adjust trim (P3):
;   Channel 3 of the ADC is measured 64 times
;   and summed up. The MSB is inverted (two's
;   complement), multiplied by cSpdMul (by
;   default 300 and cSpdAdd (by default 50)
;   is added.
;   The difference between the rClose and rOpen
;   (rClose - rOpen) is divided by byte 3 and
;   byte 2 of the result (16-by16-bit division).
;   If the 8 bit result is zero, it is set to
;   one. This is copied to register rDelta.
;
; Settings at start-up:
;   1. The stack pointer is set to RAMEND.
;   2. If any debug switches have been set,
;      those are executed.
;   3. If the push button close is pressed,
;      the T flag is cleared otherwise set.
;   4. rOpen is set to 1,550, rClose to
;      1.450 and rDelta to 1 (slow motion).
;   5. The counter TC1 is started in fast
;      PWM mode with ICR as CTC, with the
;      middle position cPwmMid (by default
;      1,500) as compare match A value, with
;      setting the OC1A output pin and clea-
;      ring OC1A on compare match and with
;      Compare Match A interrupts enabled.
;   6. The ADC is started with ADC0 as



;      first channel, with the operating
;      voltage as reference, without ADLAR
;      and with ADC interrupts enabled.
;      Sum registers are cleared and the
;      number of measuring cycles is set
;      to 64. On the three ADC channels
;      the digital drivers are disabled.
;   7. The output pina of the LED are set
;      as outputs, the LED is switched off.
;   8. The push button input pins are con-
;      figured as input pins and the pull-
;      up resistors are activted. On the
;      INT0 input interrupts on falling
;      edges are enabled. On the PCINT7
;      input the mask bit 7 is set and
;      the PCINT0 interrupt is enabled.
;   9. Interrupts are enabled and sleep
;      mode idle is activated.
;
; Interrupt processing:
;   The controller is sent to sleep. All
;   further action is executed on interrupt
;   events and after waking up the control-
;   ler following interrupt service execu-
;   tion:
;   - Compare match A interrupts set the
;     bTC1 flag, after wake-up this triggers
;     executing gate movements and operate
;     LED properties as described above.
;   - External interrupts INT0 and PCINT0
;     change the T flag accordingly (PCINT0
;     only on falling edges).
;   - ADC ready interrupts add the result
;     to the sum and restart the conversion,
;     if less than 64 measurements on the
;     same channel have been performed.
;     Otherwise the bAdc flag is set and
;     further processing of the sum result
;     is performed after wake-up.
;
; **********************************
;       R E G I S T E R S
; **********************************
;
; Used: R2:R1:R0 as 24 bit for Multipl./Div.
; Used: R5:R4:R3 as 24 bit for Multipl./Div.
; Free: R6 to R9
.def rBlink = R10 ; Blink counter
.def rPost = R11 ; Post period until inactive
.def rDelta = R12 ; Delta speed to add/subtract
.def rAdcL = R13 ; ADC sum LSB
.def rAdcH = R14 ; dto., MSB
.def rSreg = R15 ; Save/Restore status port
.def rmp = R16 ; Define multipurpose register
.def rimp = R17 ; Define multipurpose inside ints
.def rFlag = R18 ; Flag register
  .equ bAdc = 0 ; ADC conversion complete flag
  .equ bTC1 = 1 ; TC1 compare match int occurred
.def rAdcC = R19 ; ADC counter
.def rOpenL = R20 ; Open position, us
.def rOpenH = R21
.def rCloseL = R22 ; Close position, us
.def rCloseH = R23
; free: R24 to R29
; used: R31:R30 = Z for diverse purposes outside ints
;
; **********************************
;           S R A M
; **********************************
;
.dseg
.org SRAM_START
; (SRAM used only for stack)
;
; **********************************
;     C O D E  - S E G M E N T
; **********************************
;
.cseg
.org 000000



;
; **********************************
; R E S E T  &  I N T - V E C T O R S
; **********************************
  rjmp Main ; Reset vector
  rjmp Int0Isr ; EXT_INT0, push button close gate
  rjmp Pcint0Isr ; PCI0, push button open gate
  reti ; PCI1, unused
  reti ; WATCHDOG, unused
  reti ; ICP1, unused
  rjmp OC1AIsr ; OC1A
  reti ; OC1B, unused
  reti ; TC1OVF, unused
  reti ; OC0A, unused
  reti ; OC0B, unused
  reti ; OVF0, unused
  reti ; ACI, unused
  rjmp AdcIsr ; ADCC
  reti ; ERDY, unused
  reti ; USI_STR, unused
  reti ; USI_OVF, unused
;
; **********************************
;  I N T - S E R V I C E   R O U T .
; **********************************
;
; INT0 interrupt
; Falling edge on push button close input
; move gate downwards
Int0Isr:
  clt ; Downward direction, close gate
  reti
;
; PCINT0 interrupt
; Any change on the PCINT7 push button up input
Pcint0Isr:
  sbis pBtnUI,bBtnUI ; Skip if bush button up input is high
  set ; Upward direction, open gate
  reti
;
; OC1AIsr Compare Match A Int
; Occurs once in every PWM cycle (default: 20 ms)
; after compare match A has occurred, triggers
; position control of the gate (movement, post
; period and LED control)
OC1AIsr:
  in rSreg,SREG ; Save SREG
  sbr rFlag,1<<bTC1 ; Set flag
  out SREG,rSreg ; Restore SREG
  reti
;
; ADC interrupt
; AD conversion complete, add result,
; count measurements down and restart
; next conversion or set bAdc flag
AdcIsr:
  in rSreg,SREG ; Save SREG
  in rimp,ADCL ; Read LSB result
  add rAdcL,rimp ; Add to sum LSB
  in rimp,ADCH ; Read MSB result
  adc rAdcH,rimp ; Add to sum MSB with carry
  dec rAdcC ; Decrease counter
  breq AdcIsrEnd ; Channel cycle ended
  ldi rimp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
  out ADCSRA,rimp ; Start next conversion
  out SREG,rSreg ; Restore SREG
  reti
AdcIsrEnd:
  sbr rFlag,1<<bAdc ; Set ADC flag
  out SREG,rSreg ; Restore SREG
  reti
;
; **********************************
;  M A I N   P R O G R A M   I N I T
; **********************************
;
Main:
  ; Init the stack for interrupts/calls
  ldi rmp,Low(RAMEND)
  out SPL,rmp ; Init LSB stack pointer



  ; Debug switches
  .if DebugAdc == 1 ; Debug the ADC routines
    ldi rmp,cAdcChannel ; Channel (0..2)
    out ADMUX,rmp
    ldi rmp,High(cAdcResult)
    mov rAdcH,rmp
    ldi rmp,Low(cAdcResult)
    mov rAdcL,rmp
    ldi rmp,High(cOpen)
    mov rOpenH,rmp
    ldi rmp,Low(cOpen)
    mov rOpenL,rmp
    ldi rmp,High(cClose)
    mov rCloseH,rmp
    ldi rmp,Low(cClose)
    mov rCloseL,rmp
    rcall AdcRdy
    DebugLoop:
      rjmp DebugLoop
    .endif
  ; Init LED output pins
  sbi pLedD,bLedAnRdD ; LED pins as outputs
  sbi pLedD,bLEDAnGnD
  sbi pLedO,bLedAnRdO ; Set red LED on
  cbi pLedO,bLedAnGnO ; Set green LED off
  ; Init PWM output pin
  cbi pPwmO,bPwmO ; Clear PWM output port pin
  sbi pPwmD,bPwmD ; Set PWM output pin as output
  ; Init push button input pins
  cbi pBtnDD,bBtnDD ; Push button down as input
  sbi pBtnDO,bBtnDO ; Push button down pull-up on
  cbi pBtnUD,bBtnUD ; Push button up as input
  sbi pBtnUO,bBtnUO ; Push button up pull-up on
  ; If down button pressed on startup: Set flag for gate down
  set ; Upwards by default
  sbis pBtnDI,bBtnDI ; Skip if push button down input is set
  clt ; Downwards with button down pressed during start-up
  ; Set flags clear
  clr rFlag
  ; Set start parameters
  ldi rmp,High(cPwmOpenDef) ; Slightly upwards on startup
  mov rOpenH,rmp
  ldi rmp,Low(cPwmOpenDef)
  mov rOpenL,rmp
  ldi rmp,High(cPwmCloseDef) ; Slightly downwards on startup
  mov rCloseH,rmp
  ldi rmp,Low(cPwmCloseDef)
  mov rCloseL,rmp
  ldi rmp,cPwmSpeedDef ; Slowest movement possible
  mov rDelta,rmp
  ; Init TC1 as PWM
  ldi rmp,High(cPwm) ; Set the ICR, MSB
  out ICR1H,rmp
  ldi rmp,Low(cPwm) ; dto., LSB
  out ICR1L,rmp
  ldi rmp,High(cPwmMid) ; Drive motor to middle position
  out OCR1AH,rmp
  ldi rmp,Low(cPwmMid)
  out OCR1AL,rmp
  ldi rmp,(1<<COM1A1)|(1<<WGM11) ; Clear on match A, Fast PWM with ICR
  out TCCR1A,rmp
  ldi rmp,(1<<WGM13)|(1<<WGM12)|(1<<CS10) ; Prescaler=1, Fast PWM with ICR
  out TCCR1B,rmp
  ldi rmp,1<<OCIE1A ; Enable interrupts on compare match
  out TIMSK1,rmp
  ; Init ADC
  clr rAdcL ; Clear sum result
  clr rAdcH
  ldi rAdcC,64 ; 64 measurements
  ldi rmp,(1<<ADC2D)|(1<<ADC1D)|(1<<ADC0D) ; Disable input buffers
  out DIDR0,rmp
  ldi rmp,0 ; Start with channel 0, ref = operating voltage
  out ADMUX,rmp
  out ADCSRB,rmp
  ; Start first conversion
  ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
  out ADCSRA,rmp
  ; Sleep mode idle, external interrupts INT0 on falling edges
  ldi rmp,(1<<SE)|(1<<ISC01)
  out MCUCR,rmp



  ldi rmp,1<<PCINT7 ; PCINT on pin PA7
  out PCMSK0,rmp
  ldi rmp,(1<<INT0)|(1<<PCIE0) ; Enable INT0 and PCINT0 interrupts
  out GIMSK,rmp
  ; Enable interrupts
  sei ; Enable interrupts
;
; **********************************
;    P R O G R A M   L O O P
; **********************************
;
Loop:
  sleep ; Go to sleep
  nop ; Dummy on wakeup
  sbrc rFlag,bTC1 ; Skip next if bTC1 flag clear
  rcall TC1Ctrl ; Handle TC1 interrupt
  sbrc rFlag,bAdc ; Skip if bADC flag clear
  rcall AdcRdy ; Handle end of ADC cycle
  rjmp loop ; Back to sleep
;
; Handle TC1 interrupt
; Once in each PWM cycle following compare match A
TC1Ctrl:
  cbr rFlag,1<<bTC1 ; Clear flag
.if DebugPosition == 1 ; Follow positions up or down
  ret
  .endif
  in ZL,OCR1AL ; Read current position to Z
  in ZH,OCR1AH
  brts TC1CtrlUp ; Upwards mode is active
  ; Moving gate downwards
  cp ZL,rCloseL ; Check LSB
  brne Tc1CtrlDwn1
  cp ZH,rCloseH ; Check MSB
  breq TC1Post ; LSB+MSB equal, post period
Tc1CtrlDwn1:
  ; Move gate downwards
  sub ZL,rDelta ; Subtract delta
  ldi rmp,0
  sbc ZH,rmp
  cp ZH,rCloseH ; MSB equal?
  brcs Tc1CtrlDwn2 ; MSB smaller, set close position
  brne Tc1SetPos ; MSB larger, set position
  ; MSB is equal
  cp ZL,rCloseL ; LSB equal?
  brcc Tc1SetPos ; LSB larger, set position
Tc1CtrlDwn2:
  mov ZL,rCloseL ; Move to close end position
  mov ZH,rCloseH
  rjmp Tc1SetPos ; Set current position to OCR1A
TC1CtrlUp:
  ; Move gate upwards
  cp ZL,rOpenL ; Compare LSB
  brne TC1CtrlUp1
  cp ZH,rOpenH ; Compare MSB
  breq TC1Post ; Reached end position
TC1CtrlUp1:
; rClose 1500 .. 800
; rOpen 1500 .. 2200
  add ZL,rDelta ; Add rDelta
  ldi rmp,0
  adc ZH,rmp
  cp ZH,rOpenH ; Compare MSB
  brcs Tc1SetPos ; MSB smaller, set position
  brne Tc1CtrlUp2 ; MSB larger, set open position
  ; MSB equal
  cp ZL,rOpenL ; Compare LSB
  brcs Tc1SetPos ; LSB smaller, set position
TC1CtrlUp2:
  mov ZL,rOpenL  ; Set open end position
  mov ZH,rOpenH
; Write the position in Z to OCR1A
TC1SetPos:
  out OCR1AH,ZH
  out OCR1AL,ZL
  ldi rmp,cPost ; Start value post period
  mov rPost,rmp ; to rPost
  in rmp,TCCR1A ; Check if COM1A1 is on
  andi rmp,1<<COM1A1 ; Check COM1A1
  brne TC1Blink ; Is active



  ldi rmp,(1<<COM1A1)|(1<<WGM11) ; Switch COM1A1 on
  out TCCR1A,rmp ; Set pin active
TC1Blink:
  ; Gate is moving, blink the red LED
  sbi pLedO,bLedAnRdO ; Red LED on
  inc rBlink
  mov rmp,rBlink
  sbrc rmp,2 ; Blink with bit 2
  cbi pLedO,bLedAnRdO ; Red LED off
  cbi pLedO,bLedAnGnO ; Green LED off
  ret
; Gate reached end position, post period
TC1Post:
  tst rPost ; Post period over?
  breq TC1Ret ; Yes, done
  dec rPost ; Count post period down
  brne TC1Blink ; Not yet over, blink red
  ; Post period over, disable OC1A pin
  ldi rmp,1<<WGM11 ; Only the WGM bits
  out TCCR1A,rmp ; Disable pin
  cbi pPwmO,bPwmO ; clear PWM pin
  brts TC1UpLed
  ; Gate in closed position, LED is red
  sbi pLedO,bLedAnRdO ; Red LED on
  cbi pLedO,bLedAnGnO ; Green LED off
.if DebugOpenClose == 1
  set
  .endif
  ret
TC1UpLed:
  cbi pLedO,bLedAnRdO ; Red LED off
  sbi pLedO,bLedAnGnO ; Green LED on
.if DebugOpenClose == 1
  clt
  .endif
TC1Ret:
  ret
;
; ADC has ended cycle for channel
AdcRdy:
  cbr rFlag,1<<bAdc ; Clear ADC flag
  in rmp,ADMUX ; Read channel
  cpi rmp,1 ; Channel 1?
  breq AdcRdy1 ; Yes
  brcc AdcRdy2 ; Channel 2
  ; Current channel = 0
  rcall MultiplyPos
  sub ZL,R0 ; Sub result from mid
  sbc ZH,R1
  mov rCloseL,ZL
  mov rCloseH,ZH
.if DebugClose == 1
  out OCR1AH,ZH
  out OCR1AL,ZL
  .endif
  ldi rmp,1 ; Continue with channel 1
  rjmp AdcRst
AdcRdy1:
  ; Current channel = 1
  rcall MultiplyPos
  add ZL,R0 ; Add result to mid
  adc ZH,R1
  mov rOpenL,ZL
  mov rOpenH,ZH
.if DebugOpen == 1
  out OCR1AH,ZH
  out OCR1AL,ZL
  .endif
  ldi rmp,2
  rjmp AdcRst
AdcRdy2:
  ; Invert MSB ADC sum
  com rAdcH
  ; Current channel = 2
  ; Convert MSB ADC to
  ;   50..350 (1 to 7 seconds time,
  ;   50..350 Compare changes
  ;   Registers: R2:R1:R0 = 300
  ;              R5:R4:R3 = Result
  ; Divide (rClose-rOpen) by result



  ;   to yield result per step
  ;   Registers: R2:R1 as divider
  ;              R5:R4:R3 as divident
  ;              rmp as result
  ; Examples:
  ;   rClose = 2000, rOpen = 1000
  ;   MSB ADC = 0xFF:
  ;     Multiplication result = 50
  ;     Division result = 1000 / 50 = 20
  ;   MSB ADC = 0x00
  ;     Multiplication result = 350
  ;     Division result = 1000 / 350 = 2
  ; Load multiplicator (default 300)
  ldi rmp,LOW(cSpdMul)
  mov R0,rmp
  ldi rmp,HIGH(cSpdMul)
  mov R1,rmp
  clr R2
  ; Load result with adder
  clr R3
  ldi rmp,Low(cSpdAdd)
  mov R4,rmp
  ldi rmp,High(cSpdAdd)
  mov R5,rmp
; Multiply rAdcH with R1:R0
;   result in R5:R4:R3
AdcRdy2a:
  lsr rAdcH ; Shift lowest bit to carry
  brcc AdcRdy2b ; 0, do not add
  add R3,R0 ; Add the current shift to result
  adc R4,R1
  adc R5,R2
AdcRdy2b:
  lsl R0 ; Shift multiplicator one left
  rol R1
  rol R2
  tst rAdcH ; Check multiplication done?
  brne AdcRdy2a ; No
  lsl R3 ; LSB of result for rounding
  brcc AdcRdy2c
  inc R4 ; Round LSB up
  brne AdcRdy2c
  inc R5 ; Round MSB up
  ; Multiplication result is in R5:R4
AdcRdy2c:
  ; calculate difference between max and min
  ; in R2:R1:R0
  mov R1,rOpenH
  mov R0,rOpenL
  sub R0,rCloseL
  sbc R1,rCloseH
  clr R2
  ; Divide difference in R1:R0 by R5:R4
  ; rmp is 8 bit result
  ldi rmp,1
AdcRdy2d:
  lsl R0 ; Shift divident left
  rol R1
  rol R2
  brcs AdcRdy2e ; If carry, shift a one into result
  cp R1,R4 ; Compare with divisor
  cpc R2,R5
  brcc AdcRdy2e ; Shift 1 into result
  clc ; Shift 0 into result
  rjmp AdcRdy2f
AdcRdy2e:
  sub R1,R4 ; Subract divisor
  sbc R2,R5
  sec ; Shift one into the result
AdcRdy2f:
  rol rmp
  brcc AdcRdy2d
  lsl R0 ; Shift divident left for rounding
  rol R1
  rol R2
  brcs AdcRdy2g ; If a one rolls out: round up
  cp R1,R4 ; Compare LSB
  cpc R2,R5 ; and MSB
  brcs AdcRdy2h ; Smaller, do not round up
AdcRdy2g:



  inc rmp ; Round up
AdcRdy2h:
  tst rmp ; Check if result is zero
  brne AdcRdy2i
  inc rmp ; Add one to zero in rmp
AdcRdy2i:
  mov rDelta,rmp ; Copy result to rDelta
  ldi rmp,0 ; Restart with channel 0
AdcRst:
  out ADMUX,rmp
  clr rAdcL
  clr rAdcH
  ldi rAdcC,64
  ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
  out ADCSRA,rmp
  ret
;
; Multiplication of MSB ADC result
MultiplyPos:
  ldi rmp,Low(cPwmDelta) ; default: 500, to R2:R1:R0
  mov R0,rmp
  ldi rmp,High(cPwmDelta)
  mov R1,rmp
  clr R2
  clr R3 ; Clear result
  clr R4
  clr R5
MultiplyPos1:
  lsr rAdcH ; Shift bit to carry
  brcc MultiplyPos2 ; Not one
  add R3,R0
  adc R4,R1
  adc R5,R2
MultiplyPos2:
  lsl R0 ; Multiply by two
  rol R1
  rol R2
  tst rAdcH ; Check all multiplied
  brne MultiplyPos1
  lsl R3 ; Round up?
  brcc MultiplyPos3
  ldi rmp,0 ; Round one up
  adc R4,rmp
  adc R5,rmp
MultiplyPos3:
  mov R0,R4 ; Copy result to R1:R0
  mov R1,R5
  ldi ZH,High(cPwmMid) ; Load mid to Z
  ldi ZL,Low(cPwmMid)
  ret
;
; End of source code

